Institute for Informatics of the Ludwig-Maximilians-Universität München
Prof. Dr. Jasmin Blanchette
Xavier Généreux
Balazs Toth

Summer semester 2024
09.10.2024

# Solution to the Second Examination in the Course
# Interactive Theorem Proving

You have **120 minutes** at your disposal. Written or electronic aids are not permitted. Carrying electronic devices, even turned off, will be considered cheating.

Write your full name and matriculation number clearly legible on this cover sheet, as well as your name in the header on each sheet. Hand in all sheets. Leave them stapled together. Use only **pens** and **neither** the color **red nor green**.

Check that you have received all the sheets. Guidelines for writing pen-and-paper proofs are given on **page 1**. Questions can be found on **pages 2–17**. There are 6 questions for a total of 100 points.

You may use the back of the sheets for auxiliary calculations. If you use the back for actual answers, clearly mark what belongs to which question and indicate in the corresponding question where all parts of your answer can be found. Cross out everything that should not be graded.

With your signature, you confirm that you are in sufficiently good health at the beginning of the examination and that you accept this examination bindingly.

## Last name:

## First name:

## Matriculation number:

## Program of study:

## □ Please check with an X *only* if the exam should be voided and not graded.

**Bitte *nur* ankreuzen, wenn die Klausur entwertet und nicht korrigiert werden soll.**

Hierby I confirm the correctness of the above information: _____

Signature

Please leave the following table blank:

| Question | 1 | 2 | 3 | 4 | 5 | 6 | $\sum$ |
|---|---|---|---|---|---|---|---|
| Points | 20 | 25 | 25 | 8 | 12 | 10 | 100 |
| Score | | | | | | | |

**Guidelines for Paper Proofs**

We expect detailed, rigorous, mathematical proofs, but we do not ask you to write Lean proofs. You are welcome to use standard mathematical notation or Lean structured commands (e.g., `assume`, `have`, `show`, `calc`). You can also use tactical proofs (e.g., `intro`, `apply`), but then please indicate some of the intermediate goals, so that we can follow the chain of reasoning.

Major proof steps, including applications of induction and invocation of the induction hypothesis, must be stated explicitly. For each case of a proof by induction, you must list the induction hypotheses assumed (if any) and the goal to be proved. Minor proof steps corresponding to `refl`, `simp`, or `linarith` need not be justified if you think they are obvious, but you should mention which key lemmas they depend on. You should be explicit whenever you use a function definition or an introduction rule for an inductive predicate.

**Solution to Question 1 (Types and Terms):** **(20 points)**

**a)** Recall the following simplified typing rules for Lean's dependent type theory:

$$\frac{}{\texttt{C} \vdash \texttt{c} : \sigma}\ \textsc{Cst} \quad \text{if } \texttt{c} \text{ is globally declared with type } \sigma$$

$$\frac{}{\texttt{C} \vdash \texttt{x} : \sigma}\ \textsc{Var} \quad \text{if } \texttt{x} : \sigma \text{ is the rightmost occurrence of } \texttt{x} \text{ in } \texttt{C}$$

$$\frac{\texttt{C} \vdash \texttt{t} : (\texttt{x} : \sigma) \to \tau[\texttt{x}] \quad \texttt{C} \vdash \texttt{u} : \sigma}{\texttt{C} \vdash \texttt{t u} : \tau[\texttt{u}]}\ \textsc{App}'$$

$$\frac{\texttt{C}, \texttt{x} : \sigma \vdash \texttt{t} : \tau[\texttt{x}]}{\texttt{C} \vdash (\texttt{fun x} : \sigma \mapsto \texttt{t}) : (\texttt{x} : \sigma) \to \tau[\texttt{x}]}\ \textsc{Fun}'$$

Let $\texttt{Vector} : (\alpha : \texttt{Type}) \to \mathbb{N} \to \texttt{Type}$ be a type constructor. Let $\texttt{a} : \mathbb{N}$, $\texttt{g} : \mathbb{N} \to \mathbb{N} \to \mathbb{Q}$, and $\texttt{f} : (\texttt{x} : \mathbb{N}) \to \texttt{Vector}\ \mathbb{N}\ \texttt{x}$ be globally declared constants. What is the type of the following two Lean terms? Give in each case a typing derivation as justification for the type.

(i) $\texttt{f a}$ (6 points)

**PROPOSED SOLUTION:** The type is $\texttt{Vector}\ \mathbb{N}\ \texttt{a}$. The typing derivation is

$$\frac{\dfrac{}{\vdash \texttt{f} : (\texttt{x} : \mathbb{N}) \to \texttt{Vector}\ \mathbb{N}\ \texttt{x}}\ \textsc{Cst} \quad \dfrac{}{\vdash \texttt{a} : \mathbb{N}}\ \textsc{Cst}}{\vdash \texttt{f a} : \texttt{Vector}\ \mathbb{N}\ \texttt{a}}\ \textsc{App}'$$

- 2 points for the type
- 4 points for the derivation tree

(ii) $\texttt{fun x} \mapsto \texttt{g x x}$ (8 points)

**PROPOSED SOLUTION:** The type is $\mathbb{N} \to \mathbb{Q}$. The typing derivation is

$$\frac{\dfrac{\dfrac{}{\texttt{x} : \mathbb{N} \vdash \texttt{g} : \mathbb{N} \to \mathbb{N} \to \mathbb{Q}}\ \textsc{Cst} \quad \dfrac{}{\texttt{x} : \mathbb{N} \vdash \texttt{x} : \mathbb{N}}\ \textsc{Var}}{\texttt{x} : \mathbb{N} \vdash \texttt{g x} : \mathbb{N} \to \mathbb{Q}}\ \textsc{App}' \quad \dfrac{}{\texttt{x} : \mathbb{N} \vdash \texttt{x} : \mathbb{N}}\ \textsc{Var}}{\texttt{x} : \mathbb{N} \vdash \texttt{g x x} : \mathbb{Q}}\ \textsc{App}'$$

$$\frac{\texttt{x} : \mathbb{N} \vdash \texttt{g x x} : \mathbb{Q}}{\vdash (\texttt{fun x} \mapsto \texttt{g x x}) : \mathbb{N} \to \mathbb{Q}}\ \textsc{Fun}'$$

- 2 points for the type
- 6 points for the derivation tree

**b)** Let $\alpha$, $\beta$, and $\gamma$ be Lean types. Give an inhabitant for each of the following types:

- $\alpha \to \mathbb{N}$ (2 points)

**PROPOSED SOLUTION:** `fun a ↦ 0`

- 1 point for `fun a`
- 1 point for `0`

- $\alpha \to (\alpha \to \beta \to \gamma) \to \beta \to \gamma$ (2 points)

**PROPOSED SOLUTION:** `fun a f b ↦ f a b`

- 1 point for `fun a f b`
- 1 point for `f a b`

- $((\alpha \to \alpha) \to \gamma \to \alpha) \to \beta \to \gamma \to \alpha$ (2 points)

**PROPOSED SOLUTION:** `fun f b c ↦ f (fun a ↦ a) c`

- 1 point for `fun f b c`
- 1 point for `f (fun a ↦ a) c`

**Solution to Question 2 (Functional Programming):** **(25 points)**

**a)** Consider the following Lean function definition:

```
def stutter {α : Type} : List α → List α
  | []      => []
  | a :: as => a :: a :: stutter as
```

(i) Give the value of `stutter [1, 2, 3]`. (There is no need to provide intermediate steps.) (2 points)

**PROPOSED SOLUTION:** `[1, 1, 2, 2, 3, 3]`.

2 points for right answer, 0 otherwise

(ii) Prove the following Lean theorem. Make sure to follow the proof guidelines given on page 1. (8 points)

```
theorem stutter_append {α : Type} (xs ys : List α) :
  stutter (xs ++ ys) = stutter xs ++ stutter ys
```

**PROPOSED SOLUTION:**

The proof is by structural induction on `xs`.

The base case is

`stutter ([] ++ ys) = stutter [] ++ stutter ys`

Both sides simplify to `stutter ys` and are hence equal.

The induction step is

`stutter (x :: xs' ++ ys) = stutter (x :: xs') ++ stutter ys`

The induction hypothesis is

`stutter (xs' ++ ys) = stutter xs' ++ stutter ys`

The goal's left-hand side can be rewritten to `stutter (x :: (xs' ++ ys))` using basic list properties. By definition of `stutter`, this is equal to `x :: x :: stutter (xs' ++ ys)`. By the induction hypothesis, this is equal to `x :: x :: (stutter xs' ++ stutter ys)`. By definition of `stutter`, the goal's right-hand side simplifies to the same term. The two sides are equal.

- 1 point for "by (structural) induction"
- 1 point for "on `xs`"
- 1 point for statement of base case
- 1 point for proof of base case
- 1 point for statement of induction step
- 1 point for statement of induction hypothesis
- 2 points for proof of induction step (IH and `simp`)

(iii) Prove the following Lean theorem. Make sure to follow the proof guidelines given on page 1. (8 points)

```
theorem length_stutter {α : Type} (xs : List α) :
  List.length (stutter xs) = 2 * List.length xs
```

**PROPOSED SOLUTION:**

The proof is by structural induction on `xs`.

The base case is

`List.length (stutter []) = 2 * List.length []`

Both sides simplify to `0` and are hence equal.

The induction step is

`List.length (stutter (x :: xs')) = 2 * List.length (x :: xs')`

The induction hypothesis is

`List.length (stutter xs') = 2 * List.length xs'`

The induction step simplifies to

`List.length (stutter xs') + 2 = 2 * List.length xs' + 2`

By the induction hypothesis and some basic arithmetic reasoning, the two sides are equal.

- 1 point for "by (structural) induction"
- 1 point for "on `xs`"
- 1 point for statement of base case
- 1 point for proof of base case
- 1 point for statement of induction step
- 1 point for statement of induction hypothesis
- 2 points for proof of induction step (IH and `arith`)

**b)** Define a Lean function `separate` that takes a term and a list and that returns a list where the term is inserted between each pair of adjacent elements of the list. For example, `separate 0 [1, 2, 3]` should evaluate to `[1, 0, 2, 0, 3]`. (7 points)

**PROPOSED SOLUTION:**

```
def separate {α : Type} (a : α) : List α → List α
  | []              => []
  | [b]             => [b]
  | b₁ :: b₂ :: bs' => [b₁, a] ++ separate a (b₂ :: bs')
```

- 1 point for "`def separate {α : Type} (a : α)`"
- 1 point for type
- 1 point for first equation
- 1 point for second equation
- 1 point for LHS of third equation
- 2 points for RHS of third equation

**Solution to Question 3 (Inductive Predicates): (25 points)**

**a)** A binary tree is *linear* if all its nodes have at most one child. Consider the following Lean inductive predicate, which determines whether a binary tree is linear:

```
inductive Linear {α : Type} : Tree α → Prop
  | nil :
    Linear Tree.nil
  | left (a : α) (l : Tree α) (hl : Linear l) :
    Linear (Tree.node a l Tree.nil)
  | right (a : α) (r : Tree α) (hr : Linear r) :
    Linear (Tree.node a Tree.nil r)
```

Recall the `mirror` function on binary trees:

```
def mirror {α : Type} : Tree α → Tree α
  | Tree.nil        => Tree.nil
  | Tree.node a l r => Tree.node a (mirror r) (mirror l)
```

Prove the following theorem about the interaction between `Linear` and `mirror`. Make sure to follow the proof guidelines given on page 1. (9 points)

```
theorem Linear_mirror {α : Type} (t : Tree α) (ht : Linear t) :
  Linear (mirror t)
```

**PROPOSED SOLUTION:** The proof is by rule induction on `ht`.

In the `nil` case, the goal is

```
Linear (mirror Tree.nil)
```

This simplifies to `Linear Tree.nil`, which is provable using `Linear.nil`.

In the `left` case, the goal is

```
Linear (mirror (Tree.node a l Tree.nil))
```

The induction hypothesis is

```
Linear (mirror l)
```

The goal simplifies to

```
Linear (Tree.node a Tree.nil (mirror l))
```

We prove it using `Linear.right _ _ ih`, where `ih` is the induction hypothesis.

The `right` case is analogous.

- 1 point for "by (rule) induction"
- 1 point for "on `ht`"
- 1 point for statement of `nil` case
- 1 point for proof of `nil` case

- 1 point for statement of `left` or `right` case
- 1 point for the statement of the `left` or `right` induction hypothesis
- 3 points for the proofs of `left` or `right` case (IH, `simp`, and `Linear.left/right`)

**b)** Define an inductive predicate `Prefix` in Lean that takes two lists over a polymorphic type $\alpha$ as arguments and that holds when the first list is a prefix of the second. For exemple, `Prefix [1, 2] [1, 2]` and `Prefix [1, 2] [1, 2, 4]` should hold. (8 points)

**PROPOSED SOLUTION:**

```
inductive Prefix {α : Type} : List α → List α → Prop
  | nil (as : List α) :
    Prefix [] as
  | cons (a : α) (as bs : List α) (hp : Prefix as bs) :
    Prefix (a :: as) (a :: bs)
```

- 1 point for "`inductive Prefix {α : Type}`"
- 1 point for type
- 1 point for LHS of first introduction rule
- 1 point for RHS of first introduction rule
- 2 points for LHS of second introduction rule
- 2 points for RHS of second introduction rule

**c)** Define an inductive predicate `ListOver` in Lean that takes a set over a polymorphic type $\alpha$ and a list over $\alpha$ as arguments and that holds if all the elements of the list belong to the set. For example, `ListOver {4, 7} [7, 4, 7]` should hold, whereas `ListOver {7} [6]` should not hold. You may use the predicate $\in$ to test for set membership. (8 points)

**PROPOSED SOLUTION:**

```
inductive ListOver {α : Type} (A : Set α) : List α → Prop
  | nil :
    ListOver A []
  | cons (a : α) (as : List α) (ha : a ∈ A) (has : ListOver A as) :
    ListOver A (a :: as)
```

- 1 point for "`inductive ListOver {α : Type} (A : Set α)`"
- 1 point for type
- 1 point for LHS of first introduction rule
- 1 point for RHS of first introduction rule
- 2 points for LHS of second introduction rule
- 2 points for RHS of second introduction rule

**Solution to Question 4 (Effectful Programming): (8 points)**

The writer monad is a monad that stores a list of messages to be output to the console in addition to a value of type $\alpha$. In Lean, the writer monad could be defined as follows:

```
def Writer (α : Type) := α × List String

def Writer.pure {α : Type} (a : α) : Writer α :=
  (a, [])

def Writer.bind {α β : Type} (w : Writer α) (f : α → Writer β) : Writer β :=
  match w with
  | (a, msgs) =>
    match f a with
    | (b, msgs') => (b, msgs ++ msgs')

instance Writer.Pure : Pure Writer :=
  { pure := Writer.pure }

instance Writer.Bind : Bind Writer :=
  { bind := Writer.bind }
```

In the definition of `Writer.bind`, the purpose of `match` is to unfold the `Writer` tuple. It can be useful to have an equivalent representation using `Prod.fst` and `Prod.snd`:

```
lemma Writer.bind_prod {α β : Type} (w : Writer α) (f : α → Writer β) :
  w >>= f = (Prod.fst (f (Prod.fst w)), Prod.snd w ++ Prod.snd (f (Prod.fst w))) :=
  by rfl
```

**a)** In addition to `pure` and `bind`, the writer monad should offer the possibility to output messages and to retrieve the list of output messages. Implement the following two Lean functions: (3 points)

```
def Writer.getOutput {α : Type} (w : Writer α) : List String :=
```

```
def Writer.output {α : Type} (w : Writer α) (msg : String) : Writer α :=
```

**PROPOSED SOLUTION:**

```
def Writer.getOutput {α : Type} (w : Writer α) : List String :=
  match w with
  | (a, msgs) => msgs

def Writer.output {α : Type} (w : Writer α) (msg : String) : Writer α :=
  match w with
  | (a, msgs) => (a, msgs ++ [msg])
```

- 1 point for `Writer.getOutput`'s body
- 2 points for `Writer.output`'s body

**b)** Prove the following law about writer monads. Make sure to follow the proof guidelines given on page 1. In addition, give detailed steps when unfolding the definition of monad operators.

Hint: The `Writer.bind_prod` lemma presented above may be useful. (5 points)

```
theorem pure_bind {α β : Type} (a : α) (f : α → Writer β) :
  (pure a >>= f) = f a :=
```

**PROPOSED SOLUTION:** Unfolding the `bind` operator using `Writer.bind_prod` reveals

```
f a = (Prod.fst (f (Prod.fst (pure a))), Prod.snd (pure a)
      ++ Prod.snd (f (Prod.fst (pure a))))
```

Since `pure a = (a, [])`, we have `Prod.fst (pure a) = a` and `Prod.snd (pure a) = []`.

This means that

```
f a = (Prod.fst (f a), [] ++ Prod.snd (f a))
```

which are of course equal since `[]` is the identity of append.

- 2 points for unfolding `bind` correctly
- 1 point for `Prod.fst (pure a) = a`
- 1 point for `Prod.snd (pure a) = []`
- 1 point for `[]` is the identity of append.

**Solution to Question 5 (Denotational Semantics): (12 points)**

The IF language consists of all the same constructs as the WHILE language except that it does not include a `while` loop:

```
inductive Stmt : Type where
  | skip       : Stmt
  | assign     : String → (State → ℕ) → Stmt
  | seq        : Stmt → Stmt → Stmt
  | ifThenElse : (State → Prop) → Stmt → Stmt → Stmt
```

The denotational semantics of the IF language is given below:

```
def denote : Stmt → Set (State × State)
  | Stmt.skip             => Id
  | Stmt.assign x a       =>
    {st | Prod.snd st = (Prod.fst st)[x ↦ a (Prod.fst st)]}
  | Stmt.seq S T          => denote S ○ denote T
  | Stmt.ifThenElse B S T => (denote S ⨽ B) ∪ (denote T ⨽ (fun s ↦ ¬ B s))
```

where

```
Id      := {ab | Prod.snd ab = Prod.fst ab}
r ⨽ P   := {ab | ab ∈ r ∧ P (Prod.fst ab)}
r₁ ○ r₂ := {ac | ∃b, (Prod.fst ac, b) ∈ r₁ ∧ (b, Prod.snd ac) ∈ r₂}
```

We also define ⟦ ⟧ as syntactic sugar for `denote`.

**a)** Prove the following equality:

```
⟦Stmt.ifThenElse (fun s ↦ s "x" > 0)
   (Stmt.assign "x" (fun s ↦ s "x" + 1))
   Stmt.skip⟧
=
⟦Stmt.ifThenElse (fun s ↦ s "x" = 0)
   Stmt.skip
   (Stmt.assign "x" (fun s ↦ s "x" + 1))⟧
```

Recall that program variables range over natural numbers. (6 points)

**PROPOSED SOLUTION:** First, on the LHS of the equality, we unfold the semantics of the `if` statements:

```
(⟦Stmt.assign "x" (fun s ↦ s "x" + 1)⟧ ⨽ (fun s => s "x" > 0))
∪ (⟦Stmt.skip⟧ ⨽ (fun s => ¬ s "x" > 0))
```

Now consider the RHS:

```
(⟦Stmt.skip⟧ ⨽ (fun s => s "x" = 0))
∪ (⟦Stmt.assign "x" (fun s ↦ s "x" + 1)⟧ ⨽ (fun s => ¬ s "x" = 0))
```

Noting that union commutes and that `¬ n = 0` is equivalent to `n > 0` for natural numbers `n`, we conclude that the two sides are equal.

- 2 points for unfolding the semantics of `if` on the LHS

- 2 points for unfolding the semantics of `if` on the RHS
- 1 point for converting between $\neg$ `n` > `0` and `n` = `0`
- 1 point for commutativity of union

**b)** The `denote` function uses tuples in `Set (State × State)`. It is called a *relational* denotational semantics, and the tuples relate prestates to poststates. Instead, since the IF language is deterministic and terminating, we could have defined the semantics as a function from prestates to poststates. Complete the following definition of a *functional* version of `denote`. (The definition uses the `Classical.propDecidable` attribute to enable noncomputable decidability on every `Prop`.) (6 points)

```
attribute [instance 0] Classical.propDecidable

noncomputable def denote : Stmt → State → State
| Stmt.skip             => fun s ↦ s
| Stmt.assign x a       =>
| Stmt.seq S T          =>
| Stmt.ifThenElse B S T =>
```

**PROPOSED SOLUTION:**

```
noncomputable def denote : Stmt → State → State
| Stmt.skip             => fun s ↦ s
| Stmt.assign x a       => fun s ↦ s[x ↦ a s]
| Stmt.seq S T          => denote T ∘ denote S
| Stmt.ifThenElse B S T => fun s ↦ if B s then denote S s else denote T s
```

- 2 points for second RHS
- 2 points for third RHS
- 2 points for fourth RHS

**Solution to Question 6 (Mathematics):** **(10 points)**

**a)** Let $\sigma$ `: Sort 0` and $\tau$ `: Type` be Lean types. Give the type of each of the following Lean terms. (4 points)

`fun x :` $\sigma \mapsto$ `x`
$\tau \to \sigma$
`Sort 4`
$\sigma \to \tau$

**PROPOSED SOLUTION:**
$\sigma \to \sigma$
`Prop` (or `Sort 0`)
`Type 4` (or `Sort 5`)
`Type` (or `Sort 1`)

- 1 point per type

**b)** We call a `NonUnitalSemiring` a type with addition, multiplication, and a 0 element and where addition is commutative and associative, multiplication is associative and left and right distributive over addition, and 0 is the additive identity.

Complete the following `class` declaration of `NonUnitalSemiring`:

```
universe u

class NonUnitalSemiring (α : Type u) : Type u where
  add           : α → α → α
  mul           : α → α → α
  zero          :
  mul_assoc     : ∀a b c, mul (mul a b) c = mul a (mul b c)
  add_assoc     :
  left_distrib  : ∀a b c, mul a (add b c) = add (mul a b) (mul a c)
  right_distrib :
  zero_add      : ∀a, add zero a = a
  add_zero      : ∀a, add a zero = a
  zero_mul      :
  mul_zero      :
  add_comm      :
```

(6 points)

**PROPOSED SOLUTION:**

```
class NonUnitalSemiring (α : Type u) : Type u where
  add           : α → α → α
  mul           : α → α → α
  zero          : α
  mul_assoc     : ∀a b c, mul (mul a b) c = mul a (mul b c)
  add_assoc     : ∀a b c, add (add a b) c = add a (add b c)
  left_distrib  : ∀a b c, mul a (add b c) = add (mul a b) (mul a c)
  right_distrib : ∀a b c, mul (add a b) c = add (mul a c) (mul b c)
  zero_add      : ∀a, add zero a = a
  add_zero      : ∀a, add a zero = a
  zero_mul      : ∀a, mul zero a = zero
  mul_zero      : ∀a, mul a zero = zero
  add_comm      : ∀a b, add a b = add b a
```

- 1 point for `zero`
- 1 point for the `add_assoc` statement
- 1 point for the `right_distrib` statement
- 1 point for the `zero_mul` statement
- 1 point for the `mul_zero` statement
- 1 point for the `add_comm` statement